Introduction to hydrodynamic stability

This notebook has been written in *Mathematica* by

Mark J. McCready
Professor and Chair of Chemical Engineering
University of Notre Dame
Notre Dame IN 46556
USA

Mark.J.McCready.1@nd.edu
http://www.nd.edu/~mjm/

Version: 3/16/98

This notebook is intended to give a first introduction to hydrodynamic instability. This includes both the physical concepts and several of useful mathematical manipulations.

There are three parts. The first discusses the concept of linear instability theory and uses a simple wave equation to demonstrate the linearization and calculation of temporal and spatial growth.

The second part derives the stability relation for a two-layer inviscid flow, the Kelvin-Helmoltz instability.

The third part shows how to derive the basic equation of hydrodynamic stability for Newtonian fluids, the Orr-Sommerfeld equation.

The general idea of flow instability

It has been observed in nature, that the steady state solutions for different systems can become unstable to infinitesimal disturbances which should be expected to always be present, (the ground is always vibrating, buildings breath and bend, etc. ...) and possibly because of molecular motions. A common example is the formation of waves on bodies of water owing to the action of wind. The "Taylor- Couette Flow" instability is a popular laboratory instabilility that arises due to centrifugal force, and Rayleigh-Benard convection, which arises because of density differences is important both in nature and in laboratories.

Each of these instabilities has a precise, although not necessarily well understood, physical mechanism. The common feature of an instability is that infinitesimal velocity or density perturbances are amplified(by the base flow or global forces) and thus grow to finite size. Growth of distrubances could be algebraic or exponential. Typical analysis (such as those shown below) assume an exponential growth because it is expected that this would overwhelm any algebraic growth. However, algebraic analyses have been used in some situations where exponential models did not match data. It is not clear that these have matched any better, but this discussion is beyond the point of this introductory module.

Infinitesimal perturbations are expected to be in the form of noise. The question is, how to represent this when we want to model instability. Fortunately, the noise is infinitesimal, which means its amplitude is small compared to any length scale such as its wavelength. This allows the nonlinear governing equations to be well-approximated by linearized versions. The linear equations are amenable to a Fourier mode analysis that can be used to represent any noice signal as a linear combination of independent modes.

If we assume exponential growth (because is the strongest possible and what is observed in nature), and if the growth is in time (which could be how it occurs) an equation for the amplitude, *a*, of some disturbance, *a* is *a* = *a*0 Exp[ω *t*], where *a*0 is the initial amplitude of the disturbance and ω is the temporal growth rate. Even is *a*0 is of molecular dimensions (i.e. 10^-10 cm) , and the growth rate is a very reasonable (for common systems), ω = 1 / s. We find that it would take only 23 seconds for the disturbance to reach an amplitude of 1 cm !!! We thus expect a linearly unstable flow to show some evidence of growing disturbances, unless the residence time is short.

Here is the time calculation

$$\mathbf{N}\left[\mathbf{Log}\left[\frac{1}{\frac{1}{10^{10}}}\right]\right]$$

23.0259

Note that if the amplitude is growing exponentially, at some point nonlinear processes will become important. Nonlinear analysis is beyond the scope of this module.

Analysis of instability

To do mathematical analysis of an instability, we need to chose a "basestate" that is the base flow in the absence of an instability. This could be 0 velocity or it could be a falling film with no waves or a stratified flow with no waves, etc..

Since we expect that linear equations should govern the initial growth of instability, we will linearize the complete governing equations around the base flow. This is done by taking the baseflow, say $u0$ and allowing a small perturbation, say ε. Thus the complete velocity field would be $u = u0 + \varepsilon\, u1$. Note this this is for a one dimensional problem. For higher dimensions, you would have order ε components in the other directions even if the base flow in that direction was 0.

The magnitude of ε is small (<<1) because it is, for example, the amplitude to wavelength of the noise signal.

We proceed by substituting $u = u0 + \varepsilon\, u1$into the governing equations and the boundary conditions and collecting powers of ε. Because we desire the analysis to be valid for any arbitrary ε, we can separate the system into powers of ε.
The ε^0 equations will be the equations for the base state and should be identically 0.
The ε^1 equations, should give the behavior of the very small amplitude disturbance and will contain $u0$, which we know and $u1$, the distrubance that we wish to study. These equations are necessarily linear (we just linearized them with this procedure). Any higher powers of ε will be ignored and saved for when we want to do nonlinear analysis (not today!!).

To determine the response of the equation to an arbitray noise signal, we choose a mode that represents the kind of disturbance that we expect to see. If the domain is fixed and there is no flow through, (e.g., a solid beam), we might expect to use fixed spatially periodic modes that grow in time. For waves on water, we would use (traveling) spatially and temporally periodic disturbances that could grow in space and/or time.

Since the system is linear, we can examine the response of any separate mode without worrying about the effect of other modes. This linearity allows us to decompose an arbitrary disturbance into an integral (i.e. a sum) of Fourier modes, each of which will ultimately satisfy the equations and boundary conditions. By scanning the entire frequency or wavenumber range, we can be sure that we understand the effect of any initial (infintesimal) disturbance.

Example: Waves on a falling liquid film

Problem set up

For a system of traveling waves on a liquid film, the disturbance might be conveniently represented by:

$u1 = a \, \text{Exp}[i\,(k\,x - \omega\,t)]$

which is periodic in time with a circular frequency, ω [rad/s] and travels in the positive x direction with a speed, Re[ω/k]. Since we are considering that this disturbance might grow (the whole point of this exercise) we will allow for the possibility that k or ω is complex. Then this disturbance could growth spatially, Im [k] <0 or temporally, Im[ω]>0 depending on the circumstance of the flow.

For a simple,but nontrivial, example, we choose a (generalized) form of the Kuramoto-Sivshinsky equation which is derived for a falling film of water by (e.g. H. -C. Chang, *Chemical Eng. Sci*. 1986; Alekseenko et al., *AIChE J*, 1985) as:

$$\frac{\partial h}{\partial t} + c0\,\frac{\partial h}{\partial x} + \alpha\,h\,\frac{\partial h}{\partial x} + \beta\,\frac{d^2 h}{\partial x^2} + \gamma\,\frac{d^3 h}{\partial x^3} + \sigma\,\frac{d^4 h}{\partial x^4} == 0$$

where h is the deviation of the surface height from some steady value, t is time and x is the flow direction.
This is a "wave" equation in which all of the complexities of the flow have been reduced to a single equation for one dependent variable, the liquid height.

In this equation, all of the variables are real.

Clearly and constant value of h is a solution to this equation. We will choose
h=h0 and begin by linearizing the equation around h =h0.

The linearized equation

We choose, h = h0 + ε h1and substitute into the wave equation

```
temp1 = ∂t (h0 + ε h1[x, t]) + c0 ∂x (h0 + ε h1[x, t]) +
   α (h0 + ε h1[x, t]) ∂x (h0 + ε h1[x, t]) + β ∂{x,2} (h0 + ε h1[x, t]) +
   γ ∂{x,3} (h0 + ε h1[x, t]) + σ ∂{x,4} (h0 + ε h1[x, t])
```

$$\varepsilon\,h1^{(0,1)}(x,t) + c0\,\varepsilon\,h1^{(1,0)}(x,t) + \alpha\,\varepsilon\,(h0 + \varepsilon\,h1(x,t))\,h1^{(1,0)}(x,t) + \beta\,\varepsilon\,h1^{(2,0)}(x,t) + \gamma\,\varepsilon\,h1^{(3,0)}(x,t) + \varepsilon\,\sigma\,h1^{(4,0)}(x,t)$$

Get the separate powers of ε

```
temp2 = Collect[Expand[temp1], ε]
```

$$\alpha\,h1(x,t)\,h1^{(1,0)}(x,t)\,\varepsilon^2 + (h1^{(0,1)}(x,t) + c0\,h1^{(1,0)}(x,t) + h0\,\alpha\,h1^{(1,0)}(x,t) + \beta\,h1^{(2,0)}(x,t) + \gamma\,h1^{(3,0)}(x,t) + \sigma\,h1^{(4,0)}(x,t))\,\varepsilon$$

The powers of ε give:

```
Coefficient[temp2, ε, 0]
```

0

Which is the base state solution, any constant value of h0 is a solution.

```
stabeq = Coefficient[temp2, ε^1]
```

$h1^{(0,1)}(x, t) + c0\, h1^{(1,0)}(x, t) + h0\, \alpha\, h1^{(1,0)}(x, t) + \beta\, h1^{(2,0)}(x, t) + \gamma\, h1^{(3,0)}(x, t) + \sigma\, h1^{(4,0)}(x, t)$

Which is the linearized equation. Note that for a base state of h0, the nonlinear term contributes h0 $\alpha \frac{\partial h1}{\partial x}$.

This is the nonlinear term that is neglected

```
Coefficient[temp2, ε^2]
```

$\alpha\, h1(x, t)\, h1^{(1,0)}(x, t)$

- **Traveling wave mode expansion**

Now we substitute a traveling wave disturbance into the linearized equation. Note that the substitution is done in a way that works for any arbitrary derivative.

```
temp3 = stabeq /. {h1^(a1_,a2_)[x, t] :> ∂_{x,a1},{t,a2} (a Exp[I (k x - ω t)])}
```

$a\, e^{i(k x - t\omega)}\, \sigma\, k^4 - i\, a\, e^{i(k x - t\omega)}\, \gamma\, k^3 - a\, e^{i(k x - t\omega)}\, \beta\, k^2 + i\, a\, c0\, e^{i(k x - t\omega)}\, k + i\, a\, e^{i(k x - t\omega)}\, h0\, \alpha\, k - i\, a\, e^{i(k x - t\omega)}\, \omega$

We now divide out the original disturbance, which is possible since we used an exponential form.

```
temp4 = Cancel[Apart[Expand[temp3 / (a Exp[I (k x - ω t)])]]]
```

$k\,(\sigma\, k^3 - i\, \gamma\, k^2 - \beta\, k + i\, c0 + i\, h0\, \alpha) - i\, \omega$

The result is the complete dispersion relation for waves in this system. The relation between speed and frequency and wavenumber is given in this relation as is the linear growth.

$k\,(\sigma\, k^3 - i\, \gamma\, k^2 - \beta\, k + i\, c0 + i\, h0\, \alpha) - i\, \omega$

- **Temporal growth**

Consider first, spatially periodic disturbances that grow in time. This is the easy case (only 1 time derivative and it is a first derivative) and the one that most people naturally do. Of course waves on a falling film seem to grow with distance (!!)
For this case, k is real and ω is complex.

We solve the above equation for ω and then get a relation for ω. (It takes this format to do it)

```
omega = ω /. Solve[temp4 == 0, ω][[1]]
```

$-k\,(i\,\sigma\,k^3 + \gamma\,k^2 - i\,\beta\,k - \text{c0} - \text{h0}\,\alpha)$

```
Expand[omega]
```

$-i\,\sigma\,k^4 - \gamma\,k^3 + i\,\beta\,k^2 + \text{c0}\,k + \text{h0}\,\alpha\,k$

At this point we can see that certain terms contribute to the imaginary part of ω and others only to the real part. Let's look at the real part which is the frequency and relates to the wave speed.

The wave speed is ω/k so

```
speedx = Cancel[Apart[omega/k]]
```

$-i\,\sigma\,k^3 - \gamma\,k^2 + i\,\beta\,k + \text{c0} + \text{h0}\,\alpha$

We might as well use the canned program for Real and Imaginary variables.

```
Needs["Algebra`ReIm` "]
```

We some definitions of this form to make *Mathematica* know which variables are real

```
x /: Im[x] = 0;
t /: Im[t] = 0;
h /: Im[h] = 0;
h0 /: Im[h0] = 0;
α /: Im[α] = 0;
β /: Im[β] = 0;
c0 /: Im[c0] = 0;
k /: Im[k] = 0;
γ /: Im[γ] = 0;
σ /: Im[σ] = 0;
```

We can get the real part by:

```
Clear[h0]
```

```
realspeed = Re[speedx]
```

$-\gamma\,k^2 + \text{c0} + \alpha\,\text{Re(h0)}$

```
Plot[realspeed /. {h0 → 1.5, α → 1, c0 → 0, β → 3, γ → -3, σ → 4},
 {k, 0, 1}, AxesLabel -> {"k", " growthrate "}]
```



- Graphics -

We see that the wave speed is given by a base parameter, c0, and some kind of speed change due to the film thickness, h0 α. The wave speed is dependent upon the wavenumber and this is given by the term, $-\gamma k^2$.

The imaginary part of omega gives the growth rate

```
Im[omega]
```

$k^2 \beta - k^4 \sigma$

For β and σ positive, which they should be (σ is a surface tension parameter), we have the first term trying to cause the instability, (fluid inertia) and the second one countering it.

It is instructive to plot the growth rate. We see that at low wavenumber (long waves) the growth rate is increasing. However at higher wavenumber the stabilization becomes stronger and eventually the waves become stable

```
Plot[Im[omega /. {h0 → 1.5, α → 1, c0 → 0, β → 3, γ → -3, σ → 4}],
 {k, 0, 1}, AxesLabel -> {"k", " growthrate "}]
```



- Graphics -

So that there is a region for $k <\sim .85$ where the waves are unstable. The fastest growing wave is at $k =\sim 0.6$.

Now for a linear system, we can never produce a mode that is not initially excited. Thus the only waves that grow are those that were initially present. If the initial disturbance is noise, then all modes are present and we would expect to see the fastest growing mode dominate. However, this will not usually be a pure mode but a band of waves in the vicinity of the peak.

If the initial disturbance does not include modes near the peak, then others may dominate.

■ Spatial growth

We could have also considered spatially growing waves, which are most likely to be the physically relevant case for a falling film, by defining ω as real and then allowing for complex k. Even for this simple equation, the analysis is more complicated because the spatial derivatives are of higher order than the time derivatives.

```
temp5 = Expand[temp4]
```

$$\sigma k^4 - i \gamma k^3 - \beta k^2 + i \, \text{c0} \, k + i \, \text{h0} \, \alpha \, k - i \omega$$

```
kays = Solve[temp5 == 0, k];
```

Mathematica knows how to solve quartic equations analytically, so it can get the 4 roots.

Here is the first one:

```
Table[N[kays⟦1⟧ /. {h0 → 1.5, α → 1, c0 → 0, β → 3, γ → -3, σ → 4}],
 {ω, 0, 5, .5}]
```

$\{\{k \to -0.905685 - 0.543124\, i\}, \{k \to -0.950732 - 0.556116\, i\}, \{k \to -0.988369 - 0.56822\, i\}, \{k \to -1.02099 - 0.579393\, i\}, \{k \to -1.04996 - 0.589731\, i\}, \{k \to -1.07613 - 0.599345\, i\}, \{k \to -1.10008 - 0.608334\, i\}, \{k \to -1.12221 - 0.616781\, i\}, \{k \to -1.14284 - 0.624753\, i\}, \{k \to -1.16217 - 0.632308\, i\}, \{k \to -1.1804 - 0.639493\, i\}\}$

This root, [[1]], corresponds to a wave that is going upstream. This would not happen.

Here is the second one:

```
Table[N[kays⟦2⟧ /. {h0 → 1.5, α → 1, c0 → 0, β → 3, γ → -3, σ → 4}],
 {ω, 0, 5, .5}]
```

$\{\{k \to -5.55112 \times 10^{-17} + 0.336248\, i\}, \{k \to -0.144316 + 0.414148\, i\}, \{k \to -0.204826 + 0.484259\, i\}, \{k \to -0.243968 + 0.538514\, i\}, \{k \to -0.273359 + 0.58332\, i\}, \{k \to -0.297086 + 0.621825\, i\}, \{k \to -0.317088 + 0.655789\, i\}, \{k \to -0.33444 + 0.686303\, i\}, \{k \to -0.349807 + 0.714092\, i\}, \{k \to -0.363628 + 0.739667\, i\}, \{k \to -0.376209 + 0.763403\, i\}\}$

This roo,t [[2]], also corresponds to a wave that is going upstream. This also would not happen.

Here is the third one:

```
Table[N[kays⟦3⟧ /. {h0 → 1.5, α → 1, c0 → 0, β → 3, γ → -3, σ → 4}],
 {ω, 0, 20, 1}]
```

$\{\{k \to -5.55112 \times 10^{-17} + 0.\, i\}, \{k \to 0.425658 - 0.142705\, i\}, \{k \to 0.604137 - 0.652261\, i\}, \{k \to 0.573623 - 0.783253\, i\}, \{k \to 0.56842 - 0.875597\, i\}, \{k \to 0.570295 - 0.947764\, i\}, \{k \to 0.574925 - 1.00752\, i\}, \{k \to 0.580763 - 1.05881\, i\}, \{k \to 0.587137 - 1.10392\, i\}, \{k \to 0.59372 - 1.1443\, i\}, \{k \to 0.600342 - 1.18095\, i\}, \{k \to 0.606912 - 1.21456\, i\}, \{k \to 0.613379 - 1.24564\, i\}, \{k \to 0.619717 - 1.27459\, i\}, \{k \to 0.625912 - 1.30172\, i\}, \{k \to 0.63196 - 1.32726\, i\}, \{k \to 0.637859 - 1.3514\, i\}, \{k \to 0.643612 - 1.37432\, i\}, \{k \to 0.649223 - 1.39615\, i\}, \{k \to 0.654696 - 1.41699\, i\}, \{k \to 0.660037 - 1.43694\, i\}\}$

This root, [[3]], seems to be more and more unstable as the frequency increases, this is not physical for this system. However, it travels down stream and so we cannot ignore it. When a "unphysical mode" appears, this is cause for concern that the derivation has neglected something important or that our solution procedure has flaws in it. In this case, I think that the equation, which has been "generalized" to make it more interesting to study, is being used with parameters that are outside where it behaves well.

A more interesting consequence of this analysis is that you might find it very difficult to integrate this equation numerically!!

To proceed, consider both, [[3]], and [[4]].

Let's plot the two parts to see what happens.
Here we extract the 3rd root.

```
k3 = k /. kays[[3]]
```

We can plot the real wave speed as ω/k

```
speed3 =
 Plot[ω / Re[N[k3 /. {h0 → 1.5, α → 1, c0 → 0, β → 2, γ → -3, σ → 8}]],
 {ω, .001, 6}, PlotStyle -> Dashing[{.04, .02}]]
```



- Graphics -

The spatial growth rate is -Im[*k]*

```
growth3 =
 Plot[- Im[N[k3 /. {h0 → 1.5, α → 1, c0 → 0, β → 2, γ → -3, σ → 8}]],
 {ω, .001, 6}, PlotStyle -> Dashing[{.04, .02}]]
```



- Graphics -

Do this again for the 4th root

```
k4 = k /. kays[[4]];
```

First the speed

```
speed4 =
 Plot[ω / Re[N[k4 /. {h0 → 1.5, α → 1, c0 → 0, β → 2, γ → -3, σ → 8}]],
 {ω, .001, 6}]
```



- Graphics -

Then the growth

```
growth4 =
 Plot[- Im[N[k4 /. {h0 → 1.5, α → 1, c0 → 0, β → 2, γ → -3, σ → 8}]],
 {ω, .001, 6}]
```



- Graphics -

Here we plot the two wave speeds:

```
Show[speed3, speed4, AxesLabel -> {"ω", "wave speed"}]
```



- Graphics -

Here we plot the two growth curves.

```
twogrowth = Show[growth3, growth4, AxesLabel -> {"ω", "growth"}]
```



The lower mode, which starts out dashing and then becomes solid, is behaving more like a real wave than the top one. We do not see how a wave can have unbounded growth for very short waves when surface tension is present. The speed is better also as it more closely matches the qualitative behavior of the temporal analysis.

We will leave the issue of the "unphysical" mode because we can do nothing else with it right now (Although you could vary some of the parameters to see how it behaves.) but in a real problem, you would have to find out if arises because of the limitations of the derviation or the solution -- or if the instability is really there!!

- Graphics -

■ Relation between spatial and temporal growth

The spatial and temporal growth rates can be related by a Gaster Transformation (M. Gaster, *J. Fluid Mech*, **14**, p222, 1962). This is done by using the group velocity, $\partial\omega/\partial k$, to convert temporal to spatial growth,

spatial growth = temporal growth/ $(\partial\omega/\partial k)$

Here we go with this calculation.

```
omega
```

$-k\,(i\,\sigma\,k^3 + \gamma\,k^2 - i\,\beta\,k - \mathrm{c0} - \mathrm{h0}\,\alpha)$

The group velocity is defined to be real.

```
groupvel = Re[D[omega, k]]
```

$-3\,\gamma\,k^2 + \mathrm{c0} + \mathrm{h0}\,\alpha$

Here is a plot that shows the spatial growth obtained from the temporal growth.

```
pseudospatial = ParametricPlot[
  {Re[omega /. {h0 → 1.5, α → 1, c0 → 0, β → 2, γ → -3, σ → 8}],
   (Im[omega] / groupvel) /.
    {h0 → 1.5, α → 1, c0 → 0, β → 2, γ → -3, σ → 8}},
 {k, 0, 1}, AxesLabel -> {"frequency", "converted spatial growth"}]
```



- Graphics -

Now we show the real spatial and the converted spatial growths. They agree well for positive growth but not well as the amplitude of the negative growth gets larger.

Generally, it is possible to make this conversion for regions close to neutral stability.

```
Show[twogrowth, pseudospatial, PlotRange ->
  {{0, 4}, {-.25, .1}}]
```



- Graphics -

■ Why can we use a complex representation for a real disturbance ?

An important question could be that the disturbance that we used was not a real number. To make the disturbance a real number, we can use a0 Exp[I(*k* x - *ω* t)] + c.c. or
a0 Exp[I(*k*x - *ω*t)] + a0cc Exp[-I(*k*x - *ω* t)]. If this is substituted into the equation we will get twice as many terms. One set will have Exp[I(*k*x - *ω*t)] as a factor and the other set will have Exp[-I(*k*x - *ω*t)] as a factor. For this expression to always be true (i.e. for any x and t), these sets must each be equal to 0 separately. The analysis would then be identical to the one done above.

Note that if we are doing a nonlinear problem, we would have to keep track of the complex conjugate part as it may contain additional information.

Kelvin-Helmholtz Instability for finite depth

- **Problem set up**

A good reference for this section is R. L. Panton, Incompressible flow, Wiley, 1984

The generation of water waves by wind has puzzled and fascinated scientists for centuries. It is a classic stability problem that, as it turns out, was not really completely understood until about the 1970's.

This problem was attacked very early in the study of hydrodyanmic stability for the idealized case of inviscid flows. This solutions is known as the Kelvin-Helmholtz instability and it **does not** predict the observed behavior of water wave generation!
It does not work for any two layer system where the viscosities of the two fluids are very different, no matter how high the Reynolds numbers are. Still it is a useful instability for study as it provides a nice physical mechanism to easily understood situation.



Consider a stratified horizontal flow of two inviscid fluids. The top fluid is "2" and has a velocity of U_2. The bottom is "1" and has a velocity of U_1. The interface is at y = 0 and the top wall is at $y = h_2$, the bottom wall is at $y = -h_1$.

Since each phase is inviscid, Laplace's equation for the velocity potential can be written,

$\nabla^2\Phi_1 = 0$ and $\nabla^2\Phi_2 = 0.$

where

$U_1 = \nabla\Phi_1$ and $U_2 = \nabla\Phi_2.$

- **Laplace equation solution**

First we should find solutions for Φ in each phase that fits the necessary boundary conditions. This would be that there is no flow through the top and bottom walls.
The other boundary conditions will be introduced along the way.

The waves will be traveling in the x direction, grow and oscillate in time and must decay away from the interface in y.

For the lower phase, why don't we guess a form as

Φ_1[x,y,t] = U1 x + ϕ_1[y] Exp[I *k*(x - *c* t)]

ϕ_1[y] = B1 Cosh[k (y+h1)]

this gives the base flow plus the disturbance.

For the upper phase, how about

Φ_2[x,y,t] = U2 x + ϕ_2[y] Exp[I k(x - c t)]

ϕ_2[y] = A2 Cosh[k (y-h2)]

In these expressions, *k* is wavenumber, *c* is wave speed, x is the direction of travel and t is time.

Let's see how well we have guessed!!

For the upper phase

```
Φ1 = U1 x + B1 Exp[I k (x - c t)] Cosh[k (y + h1)]
```

$U1\, x + B1\, e^{i\, k\, (x - c\, t)} \cosh(k\, (h1 + y))$

Check in Laplace's equation and it works

```
D[Φ1, {x, 2}] + D[Φ1, {y, 2}]
```

0

Now check the lower boundary condition, it works also

```
∂y Φ1 /. y → -h1
```

0

now try the upper phase.

```
Φ2 = U2 x + A2 Exp[I k (x - c t)] Cosh[k (y - h2)]
```

$\text{U2}\, x + \text{A2}\, e^{i k (x - c t)} \cosh(k (y - \text{h2}))$

test in LaPlaces' equation

```
D[Φ2, {x, 2}] + D[Φ2, {y, 2}]
```

0

Test the boundary condition

```
∂y Φ2 /. y → h2
```

0

OK so far, so good.

- **Kinematic boundary condition**

Whenever there is an interface between fluids that has waves on it, we need a relation between the velocity field and the motion of the interface. This can formulated by using the substantial derivative of the surface position function, $\eta(x,t)$ and equating this to the normal velocity of the fluid at the interface.

$\frac{D\eta(x,t)}{Dt}$ = normal velocity at the interface = $\frac{\partial\phi(x,y,t)}{\partial y}$ (note that this potential is just for the perturbation velocity because there is no average normal velocity)

this becomes

$\partial\eta/\partial t + U\,\partial\eta/\partial x + W\,\partial\eta/\partial z = \partial\,\phi/\partial y$ @ $y=\eta$.

In this example we will not consider the transverse direction, so W = 0 and $\partial\,()/\partial z = 0$. In its present form, the second term on the right is nonlinear. We just need a linearized version for this linear stability problem. This is

$\partial\eta/\partial t + U1\,\partial\eta/\partial x = \partial\,\phi/\partial y$ @ $y=\eta$.

because the perturbation term, $\partial\phi/\partial x\,\partial\eta/\partial x$ is order a^2, which can be ignored.

There is still one problem remaining. It would be exceedingly inconvenient to evaluate this condition at $y = \eta$, since this changes with x and t. It turns out that this problem can be solved by domain perturbation. We would like to evaluate these equations at y = 0 and we need the real values of the variables at this point. We can get them by expanding the velocities in a Taylor series. The surface variable is not a function of y and thus does not require any expanding.

$U(y=h) = U(y=0) + a\,\partial U/\partial y(y=0)$ + H.O.T.

$v(y=h) = v(y=0) + a\,\partial v/\partial y\,(y=0)$ + H.O.T

The correction terms are of order a and would not enter in this linear problem. Further, for this inviscid flow, there is no velocity gradient in the y direction so there is no correction for the flow direction velocity.

As a result of this discussion, we have this "kinematic" condition for both phases related to the interfacial shape.

$\partial\eta/\partial t + U1\,\partial\eta/\partial x = \partial\,\phi 1/\partial y$ @ y= 0.

$\partial\eta/\partial t + U2\,\partial\eta/\partial x = \partial\,\phi 2/\partial y$ @ y= 0.

Here is the kinematic condition

```
kc1 = ∂y ϕ1[x, y, t] - ∂t η[x, t] - U1 ∂x η[x, t]
```

$$-\eta^{(0,1)}(x,t) - \text{U1}\,\eta^{(1,0)}(x,t) + \phi_1^{(0,1,0)}(x,y,t)$$

Here we impose the kinematic condition for phase 1 by substituting for $\phi 1$ and η. Note that $\hat{\eta}$ is just the wave amplitude.

```
temp1 = kc1 /. {η[x, t] → η̂ Exp[I k (x - c t)],
    ϕ1[x, y, t] → B1 Cosh[k (y + h1)] Exp[I k (x - c t)],
    ϕ2[x, y, t] → A2 Cosh[k (y - h2)] Exp[I k (x - c t)], η^(a1_,a2_)[x, t] :→
     ∂{x,a1},{t,a2} (η̂ Exp[I k (x - c t)]), ϕ1^(a1_,a2_,a3_)[x, y, t] :→
     ∂{x,a1},{y,a2},{t,a3} (B1 Cosh[k (y + h1)] Exp[I k (x - c t)]),
    ϕ2^(a1_,a2_,a3_)[x, y, t] :→
     ∂{x,a1},{y,a2},{t,a3} (A2 Cosh[k (y - h2)] Exp[I k (x - c t)])}
```

$i\, c\, e^{i k (x-c t)}\, k\, \hat{\eta} - i\, e^{i k (x-c t)}\, k\, \mathrm{U1}\, \hat{\eta} + \mathrm{B1}\, e^{i k (x-c t)}\, k \sinh(k\, (\mathrm{h1} + y))$

```
temp2 = Cancel[Expand[temp1 / Exp[I k (x - c t)]]] /. y → 0
```

$i\, c\, k\, \hat{\eta} - i\, k\, \mathrm{U1}\, \hat{\eta} + \mathrm{B1}\, k \sinh(\mathrm{h1}\, k)$

```
BEE1 = B1 /. Solve[temp2 == 0, B1][[1]]
```

$-\frac{\mathrm{csch}(\mathrm{h1}\, k)\, (i\, c\, k\, \hat{\eta} - i\, k\, \mathrm{U1}\, \hat{\eta})}{k}$

So now we know B1 in terms of $\hat{\eta}$.
Now do the second kinematic condition

```
kc2 = ∂y ϕ2[x, y, t] - ∂t η[x, t] - U2 ∂x η[x, t]
```

$-\eta^{(0,1)}(x, t) - \mathrm{U2}\, \eta^{(1,0)}(x, t) + \phi_2^{(0,1,0)}(x, y, t)$

```
temp3 = kc2 /. {η[x, t] → η̂ Exp[I k (x - c t)],
    ϕ1[x, y, t] → B1 Cosh[k (y + h1)] Exp[I k (x - c t)],
    ϕ2[x, y, t] → A2 Cosh[k (y - h2)] Exp[I k (x - c t)], η^(a1_,a2_)[x, t] :→
     ∂{x,a1},{t,a2} (η̂ Exp[I k (x - c t)]), ϕ1^(a1_,a2_,a3_)[x, y, t] :→
     ∂{x,a1},{y,a2},{t,a3} (B1 Cosh[k (y + h1)] Exp[I k (x - c t)]),
    ϕ2^(a1_,a2_,a3_)[x, y, t] :→
     ∂{x,a1},{y,a2},{t,a3} (A2 Cosh[k (y - h2)] Exp[I k (x - c t)])}
```

$i\, c\, e^{i k (x-c t)}\, k\, \hat{\eta} - i\, e^{i k (x-c t)}\, k\, \mathrm{U2}\, \hat{\eta} + \mathrm{A2}\, e^{i k (x-c t)}\, k \sinh(k\, (y - \mathrm{h2}))$

```
temp4 = Cancel[Expand[temp3 / Exp[I k (x - c t)]]] /. y → 0
```

$i\, c\, k\, \hat{\eta} - i\, k\, \mathrm{U2}\, \hat{\eta} - \mathrm{A2}\, k \sinh(\mathrm{h2}\, k)$

Now we can get A2

```
AY2 = A2 /. Solve[% == 0, A2][[1]]
```

$\frac{\mathrm{csch}(\mathrm{h2}\, k)\, (i\, c\, k\, \hat{\eta} - i\, k\, \mathrm{U2}\, \hat{\eta})}{k}$

- **Dynamic Boundary condition**

The pressure will not be constant along the interface, if waves are present. We can find out the variation by applying the Bernoulli equation for each phase at the interface. The pressure must be the same on each side of the interface, except for the effect of durface tension. This is a "dynamic condition" (pressure changes with velocity) and is referred to in this way.

We can write the Bernoulli equation for each phase as:

$$\partial\phi_1/\partial t + 1/2\,(\nabla\phi)^2 + P_1/\rho_1 + g\,\eta(x,t) = 0$$

$$\partial\phi_2/\partial t + 1/2\,(\nabla\phi)^2 + P_2/\rho_2 + g\,\eta(x,t) = 0$$

We will linearize the $(\nabla\phi)^2$ term. The difference in the pressure can be written as

$(P_2 - P_1) = -\gamma \frac{\partial^2 \eta(x,t)}{\partial x^2}$

where γ is the surface tension coefficient.

Here is the dynamic boundary condition.

```
dc =
ρ2 (-∂t ϕ2[x, y, t] - U2 ∂x ϕ2[x, y, t] - g η[x, t]) -
 ρ1 (-∂t ϕ1[x, y, t] - U1 ∂x ϕ1[x, y, t] - g η[x, t]) - γ ∂{x,2} η[x, t]
```

$$-\gamma\,\eta^{(2,0)}(x,t) - \rho_1\left(-g\,\eta(x,t) - \phi_1^{(0,0,1)}(x,y,t) - \text{U1}\,\phi_1^{(1,0,0)}(x,y,t)\right) + \rho_2\left(-g\,\eta(x,t) - \phi_2^{(0,0,1)}(x,y,t) - \text{U2}\,\phi_2^{(1,0,0)}(x,y,t)\right)$$

Substitute all that we know

```
temp6 = dc /. {η[x, t] → η̂ Exp[I k (x - c t)],
    ϕ1[x, y, t] → B1 Cosh[k (y + h1)] Exp[I k (x - c t)],
    ϕ2[x, y, t] → A2 Cosh[k (y - h2)] Exp[I k (x - c t)], η^(a1_,a2_)[x, t] :→
     ∂{x,a1},{t,a2} (η̂ Exp[I k (x - c t)]), ϕ1^(a1_,a2_,a3_)[x, y, t] :→
     ∂{x,a1},{y,a2},{t,a3} (B1 Cosh[k (y + h1)] Exp[I k (x - c t)]),
    ϕ2^(a1_,a2_,a3_)[x, y, t] :→
     ∂{x,a1},{y,a2},{t,a3} (A2 Cosh[k (y - h2)] Exp[I k (x - c t)])}
```

$$e^{i k (x - c t)}\,\gamma\,\hat{\eta}\,k^2 - \left(i\,\text{B1}\,c\,e^{i k (x-c t)}\,k\cosh(k\,(\text{h1}+y)) - i\,\text{B1}\,e^{i k (x-c t)}\,k\,\text{U1}\cosh(k\,(\text{h1}+y)) - e^{i k (x-c t)}\,g\,\hat{\eta}\right)\rho_1 + \left(i\,\text{A2}\,c\,e^{i k (x-c t)}\,k\cosh(k\,(y-\text{h2})) - i\,\text{A2}\,e^{i k (x-c t)}\,k\,\text{U2}\cosh(k\,(y-\text{h2})) - e^{i k (x-c t)}\,g\,\hat{\eta}\right)\rho_2$$

```
temp7 = Cancel[Expand[temp6 / Exp[I k (x - c t)]]] /. y → 0
```

$$\gamma\,\hat{\eta}\,k^2 - i\,\text{B1}\,c\cosh(\text{h1}\,k)\,\rho_1\,k + i\,\text{B1}\,\text{U1}\cosh(\text{h1}\,k)\,\rho_1\,k + i\,\text{A2}\,c\cosh(\text{h2}\,k)\,\rho_2\,k - i\,\text{A2}\,\text{U2}\cosh(\text{h2}\,k)\,\rho_2\,k + g\,\hat{\eta}\,\rho_1 - g\,\hat{\eta}\,\rho_2$$

Substitute for A2 and B1

This command substitutes for A2 from above

```
temp8 = temp7 /. A2 → AY2
```

$$\gamma \hat{\eta} k^2 - i \,\text{B1}\, c \cosh(\text{h1}\, k) \rho_1 k + i \,\text{B1 U1} \cosh(\text{h1}\, k) \rho_1 k + g \hat{\eta} \rho_1 - g \hat{\eta} \rho_2 + i c \coth(\text{h2}\, k)(i c k \hat{\eta} - i k \,\text{U2}\, \hat{\eta}) \rho_2 - i \,\text{U2} \coth(\text{h2}\, k)(i c k \hat{\eta} - i k \,\text{U2}\, \hat{\eta}) \rho_2$$

This command substitutes for B1 from above

```
temp9 = temp8 /. B1 → BEE1
```

$$\gamma \hat{\eta} k^2 + g \hat{\eta} \rho_1 + i c \coth(\text{h1}\, k)(i c k \hat{\eta} - i k \,\text{U1}\, \hat{\eta}) \rho_1 - i \,\text{U1} \coth(\text{h1}\, k)(i c k \hat{\eta} - i k \,\text{U1}\, \hat{\eta}) \rho_1 - g \hat{\eta} \rho_2 + i c \coth(\text{h2}\, k)(i c k \hat{\eta} - i k \,\text{U2}\, \hat{\eta}) \rho_2 - i \,\text{U2} \coth(\text{h2}\, k)(i c k \hat{\eta} - i k \,\text{U2}\, \hat{\eta}) \rho_2$$

```
temp10 = Expand[temp9]
```

$$-k \coth(\text{h1}\, k) \hat{\eta} \rho_1 c^2 - k \coth(\text{h2}\, k) \hat{\eta} \rho_2 c^2 + 2 k \,\text{U1} \coth(\text{h1}\, k) \hat{\eta} \rho_1 c + 2 k \,\text{U2} \coth(\text{h2}\, k) \hat{\eta} \rho_2 c + k^2 \gamma \hat{\eta} + g \hat{\eta} \rho_1 - k \,\text{U1}^2 \coth(\text{h1}\, k) \hat{\eta} \rho_1 - g \hat{\eta} \rho_2 - k \,\text{U2}^2 \coth(\text{h2}\, k) \hat{\eta} \rho_2$$

- **Dispersion relation and growth expression**

By substituting, we have obtained a solution for c in terms of known variables. Note that $\hat{\eta}$, which is arbitrary, is multiplying every term linearly and will cancel.

```
temp11 = Collect[temp10, c]
```

$$(-k \coth(\text{h1}\, k) \hat{\eta} \rho_1 - k \coth(\text{h2}\, k) \hat{\eta} \rho_2) c^2 + (2 k \,\text{U1} \coth(\text{h1}\, k) \hat{\eta} \rho_1 + 2 k \,\text{U2} \coth(\text{h2}\, k) \hat{\eta} \rho_2) c + k^2 \gamma \hat{\eta} + g \hat{\eta} \rho_1 - k \,\text{U1}^2 \coth(\text{h1}\, k) \hat{\eta} \rho_1 - g \hat{\eta} \rho_2 - k \,\text{U2}^2 \coth(\text{h2}\, k) \hat{\eta} \rho_2$$

This can be written more concisely as

```
temp12 = FullSimplify[%]
```

— *General::spell1 : Possible spelling error: new symbol name "temp12" is similar to existing symbol "temp2".*

$$\hat{\eta} (\gamma k^2 + (g - k (c - \text{U1})^2 \coth(\text{h1}\, k)) \rho_1 - (k \coth(\text{h2}\, k) (c - \text{U2})^2 + g) \rho_2)$$

This is the dispersion relation for the wave speed.
The solutions are:

```
cees = Solve[temp12 == 0, c]
```

$$\{\{c \to (-k(-2\,\text{U1}\coth(\text{h1}\,k)\rho_1 - 2\,\text{U2}\coth(\text{h2}\,k)\rho_2) - \sqrt{(k^2(-2\,\text{U1}\coth(\text{h1}\,k)\rho_1 - 2\,\text{U2}\coth(\text{h2}\,k)\rho_2)^2 - 4k(\coth(\text{h1}\,k)\rho_1 + \coth(\text{h2}\,k)\rho_2)(-\gamma k^2 + \text{U1}^2\coth(\text{h1}\,k)\rho_1 k + \text{U2}^2\coth(\text{h2}\,k)\rho_2 k - g\rho_1 + g\rho_2)))}/(2k(\coth(\text{h1}\,k)\rho_1 + \coth(\text{h2}\,k)\rho_2))\},$$

$$\{c \to (\sqrt{(k^2(-2\,\text{U1}\coth(\text{h1}\,k)\rho_1 - 2\,\text{U2}\coth(\text{h2}\,k)\rho_2)^2 - 4k(\coth(\text{h1}\,k)\rho_1 + \coth(\text{h2}\,k)\rho_2)(-\gamma k^2 + \text{U1}^2\coth(\text{h1}\,k)\rho_1 k + \text{U2}^2\coth(\text{h2}\,k)\rho_2 k - g\rho_1 + g\rho_2))} - k(-2\,\text{U1}\coth(\text{h1}\,k)\rho_1 - 2\,\text{U2}\coth(\text{h2}\,k)\rho_2))/(2k(\coth(\text{h1}\,k)\rho_1 + \coth(\text{h2}\,k)\rho_2))\}\}$$

Which should be a set of upstream and downstream traveling waves.
c1 will be the upstream wave and will not appear in a flow system
c2 will be the downstream wave that we would like to study the behavior of

```
c1 = c /. cees⟦1⟧
```

$$(-k(-2\,\text{U1}\coth(\text{h1}\,k)\rho_1 - 2\,\text{U2}\coth(\text{h2}\,k)\rho_2) - \sqrt{(k^2(-2\,\text{U1}\coth(\text{h1}\,k)\rho_1 - 2\,\text{U2}\coth(\text{h2}\,k)\rho_2)^2 - 4k(\coth(\text{h1}\,k)\rho_1 + \coth(\text{h2}\,k)\rho_2)(-\gamma k^2 + \text{U1}^2\coth(\text{h1}\,k)\rho_1 k + \text{U2}^2\coth(\text{h2}\,k)\rho_2 k - g\rho_1 + g\rho_2)))}/(2k(\coth(\text{h1}\,k)\rho_1 + \coth(\text{h2}\,k)\rho_2))$$

```
c2 = c /. cees⟦2⟧
```

$$(\sqrt{(k^2(-2\,\text{U1}\coth(\text{h1}\,k)\rho_1 - 2\,\text{U2}\coth(\text{h2}\,k)\rho_2)^2 - 4k(\coth(\text{h1}\,k)\rho_1 + \coth(\text{h2}\,k)\rho_2)(-\gamma k^2 + \text{U1}^2\coth(\text{h1}\,k)\rho_1 k + \text{U2}^2\coth(\text{h2}\,k)\rho_2 k - g\rho_1 + g\rho_2))} - k(-2\,\text{U1}\coth(\text{h1}\,k)\rho_1 - 2\,\text{U2}\coth(\text{h2}\,k)\rho_2))/(2k(\coth(\text{h1}\,k)\rho_1 + \coth(\text{h2}\,k)\rho_2))$$

- **Example calculations**

The air water system, wind is 300 cm/s and the layers are deep.   The system is stable.

```
c2numb = c2 /. {ρ2 → .00125, ρ1 → 1.020, g → 980, γ → 74, U1 → 1,
   U2 → 300, h1 → 10, h2 → 10}
```

$$\frac{1}{k}\left(0.489596\left(2.79\,k\coth(10\,k) + \sqrt{(7.7841\,k^2\coth^2(10\,k) - 4.085\,k\coth(10\,k)(-74\,k^2 + 113.52\,\coth(10\,k)\,k - 998.375))}\right)\tanh(10\,k)\right)$$

```
kh1 = Plot[Re[N[c2numb]], {k, .01, 30},
  AxesLabel → {"wavenumber", "wavespeed"}]
```



- Graphics -

This is the typical shape of a water wave dispersion curve. Gravity waves (long waves) get faster as they get longer. In contrast, capillary waves get faster as they get shorter. (These waves are linearly stable.)

Here is the growth curve which is always 0. This means that the theory predicts that no wave modes will grow. Note that for real fluids, stable waves have negative growth rates.

```
kh2 = Plot[Im[N[c2numb]], {k, .01, 30},
  AxesLabel → {"wavenumber", "wave growth"}]
```



- Graphics -

Now suppose the depth is not as large

```
c2numb2 = c2 /. {ρ2 → .00125, ρ1 → 1.020, g → 980, γ → 74, U1 → 1,
    U2 → 300, h1 → .2, h2 → .2}
```

$$\frac{1}{k}\Big(0.489596\ \big(2.79\ k\coth(0.2\ k) + \sqrt{(7.7841\ k^2\ \coth^2(0.2\ k) - 4.085\ k\coth(0.2\ k)\,(-74\ k^2 + 113.52\ \coth(0.2\ k)\,k - 998.375))}\big)\ \tanh(0.2\ k)\Big)$$

```
kh3 = Plot[Re[N[c2numb2]], {k, .01, 30},
  AxesLabel → {"wavenumber", "wavespeed"}]
```



- Graphics -

We might as well compare the wave speeds for two different depths. If the depth is too small, the long waves do not speed up.

```
Show[kh1, kh3]
```



- Graphics -

Now the growth curve which is again always 0.

```
kh4 = Plot[Im[N[c2numb2]], {k, .01, 30},
  AxesLabel → {"wavenumber", "wave growth"}]
```



- Graphics -

Now, let's turn up the gas speed

```
c2numb3 = c2 /. {ρ2 → .00125, ρ1 → 1.020, g → 980, γ → 74, U1 → 1,
   U2 → 600, h1 → .2, h2 → .2}
```

$$\frac{1}{k}\Big(0.489596\ \Big(3.54\ k \coth(0.2\ k) + \sqrt{(12.5316\ k^2 \coth^2(0.2\ k) - 4.085\ k \coth(0.2\ k)\,(-74\ k^2 + 451.02\ \coth(0.2\ k)\,k - 998.375))}\Big) \tanh(0.2\ k)\Big)$$

```
kh5 = Plot[Re[N[c2numb3]],
  {k, .01, 30}, AxesLabel → {"wavenumber", "wave speed"},
  PlotRange -> All]
```



- Graphics -

We see that the behavior of the speed in the low wavenumber region is different. This is what the model predicts when the waves become unstable. This does not seem correct and is not predicted by more accurate models.

Now the growth curve is not always 0. Wavenumbers less than about 5/cm are predicted to be unstable.

```
kh6 = Plot[Im[N[c2numb3]], {k, .01, 30},
  AxesLabel → {"wavenumber", "wave growth"}]
```



- Graphics -

You would find that the velocity difference will need to be about 660 cm/s for large depths but less if the depths are lowered and the speed is changed such as the example above.

Now make both layers deep. We need to make the velocity higher.

```
c2numb4 = c2 /. {ρ2 → .00125, ρ1 → 1.020, g → 980, γ → 74, U1 → 1,
   U2 → 670, h1 → 10, h2 → 10}
```

$$\frac{1}{k}\Big(0.489596\ \big(3.715\ k \coth(10\ k) + \sqrt{(13.8012\ k^2 \coth^2(10\ k) - 4.085\ k \coth(10\ k)(-74\ k^2 + 562.145\ \coth(10\ k)\ k - 998.375))}\big)\ \tanh(10\ k)\Big)$$

```
kh7 = Plot[Re[N[c2numb4]], {k, .01, 30},
  AxesLabel → {"wavenumber", "wave speed"}]
```



- Graphics -

We see that the behavior of the speed in the low wavenumber region is different. This is what the model predicts when the waves become unstable. This does not seem correct and is not predicted by more accurate models.

Now the growth curve is bounded away from 0 wavenumber.

```
kh8 = Plot[Im[N[c2numb4]], {k, .01, 30},
  AxesLabel → {"wavenumber", "wave growth"}]
```



- Graphics -

Which shows the temporal grow rate for an interfacial disturbance.
This calculation is easy and neat, however there is one big problem, it does not match the linear behavior for the air water system. The predicted point of instability is too high; the real value would be more like 3 m/s. Furthermore, note the speed of unstable waves. They are predicted to travel at the speed (ρ_1 u1+ρ_2 u2)/(ρ_1+ρ_2), which is not correct.

Now check the upstream wave:

```
c1numb1 = c1 /. {ρ2 → .00125, ρ1 → 1.020, g → 980, γ → 74, U1 → 1,
   U2 → 670, h1 → 10, h2 → 10}
```

$$\frac{1}{k}\Big(0.489596\ \Big(3.715\ k\coth(10\ k) - \sqrt{(13.8012\ k^2\coth^2(10\ k) - 4.085\ k\coth(10\ k)(-74\ k^2 + 562.145\ \coth(10\ k)\ k - 998.375))}\Big)\tanh(10\ k)\Big)$$

```
kh9 = Plot[Re[c1numb1], {k, .01, 20},
  AxesLabel → {"wavenumber", "wave speed"}]
```



- Graphics -

The wave speed is now usually negative

```
kh10 = Plot[Im[c1numb1], {k, .01, 20},
  AxesLabel → {"wavenumber", "wave growth"}]
```



- Graphics -

The upstream wave does have a negative growth region that corresponds to the wavenumbers where the downstream wave is predicted to be unstable.

■ Examination of the physics

■ No surface tension

It is worthwhile to get some additional physical feel for this instability, even if it is not applicable to a real air-water system. First, what if no surface tension

```
c2numb5 = c2 /. {ρ2 → .00125, ρ1 → 1.020, g → 980, γ → 0, U1 → 1,
   U2 → 670, h1 → 10, h2 → 10}
```

$$\frac{1}{k}\Big(0.489596\ \big(3.715\ k \coth(10\ k) + \sqrt{(13.8012\ k^2 \coth^2(10\ k) - 4.085\ k \coth(10\ k)\,(562.145\ k \coth(10\ k) - 998.375\,))}\big)\ \tanh(10\ k)\Big)$$

```
kh11 = Plot[Re[c2numb5], {k, .01, 20},
  AxesLabel → {"wavenumber", "wave speed"}, PlotRange → All]
```



- Graphics -

The wave speed looks peculiar. Without surface tension to stabilize the short waves, these are unstable. If they are unstable, the wave speed is just the density weighted average of the upper and lower depths.

```
kh12 = Plot[Im[c2numb5], {k, .01, 20},
  AxesLabel → {"wavenumber", "wave growth"}]
```

— *General::spell1 : Possible spelling error: new symbol name "kh12" is similar to existing symbol "kh2".*



- Graphics -

Without surface tension, all waves shorter than some cutoff are unstable.

So surface tension stabilizes the short waves, which is reasonable.

- **What is the point of neutral stability and what is the mechanism?**

Let's consider the simplest case, both layers deep, air-water (ρ_2 <<< ρ_1) and that the velocity, U1 is much less than U2.

```
nuet1 = c2 /. {h2 -> h1, U1 -> 0}
```

$$\left(2\,k\,\text{U2}\coth(\text{h1}\,k)\,\rho_2+\sqrt{\left(4\,k^2\,\text{U2}^2\coth^2(\text{h1}\,k)\,\rho_2^2-4\,k\,(\coth(\text{h1}\,k)\,\rho_1+\coth(\text{h1}\,k)\,\rho_2)\,(-\gamma\,k^2+\text{U2}^2\coth(\text{h1}\,k)\,\rho_2\,k-g\,\rho_1+g\,\rho_2)\right)}\right)/\left(2\,k\,(\coth(\text{h1}\,k)\,\rho_1+\coth(\text{h1}\,k)\,\rho_2)\right)$$

We know by definition for h1->Infinity

```
nuet2 = nuet1 /. {Coth[h1 k] -> 1}
```

$$\frac{2\,k\,\text{U2}\,\rho_2+\sqrt{4\,k^2\,\text{U2}^2\,\rho_2^2-4\,k\,(\rho_1+\rho_2)\,(-\gamma\,k^2+\text{U2}^2\,\rho_2\,k-g\,\rho_1+g\,\rho_2)}}{2\,k\,(\rho_1+\rho_2)}$$

For instability, we need the expression inside the root to be negative. Let's look at it.

```
nuet3 = Apart[nuet2]
```

$$\frac{\text{U2}\,\rho_2}{\rho_1+\rho_2}+\frac{\sqrt{4\,k^2\,\text{U2}^2\,\rho_2^2-4\,k\,(\rho_1+\rho_2)\,(-\gamma\,k^2+\text{U2}^2\,\rho_2\,k-g\,\rho_1+g\,\rho_2)}}{2\,k\,(\rho_1+\rho_2)}$$

```
nuet4 = Numerator[nuet3[[2]]][[1]]
```

$$4 k^2 \, \mathrm{U2}^2 \, \rho_2^2 - 4 k (\rho_1 + \rho_2)(-\gamma k^2 + \mathrm{U2}^2 \rho_2 k - g \rho_1 + g \rho_2)$$

```
nuet5 = Collect[Collect[ExpandAll[nuet4 / ρ1 / 4], ρ2], k]
```

$$\left(\frac{\rho_2 \gamma}{\rho_1} + \gamma\right) k^3 - \mathrm{U2}^2 \rho_2 k^2 + \left(g \rho_1 - \frac{g \rho_2^2}{\rho_1}\right) k$$

Now we see that for each power of k, how important the ρ_2 terms are. We choose the important ones by making the others 0. This is inelegant, but the easiest way to do it.

```
neutral1 = nuet5 /. {γ ρ2 -> 0, ρ2^2 -> 0}
```

$$\gamma k^3 - \mathrm{U2}^2 \rho_2 k^2 + g \rho_1 k$$

If this quantity is greater than 0, the flow is unstable. The first term is the restoring force of surface tension, the third term is the restoring force of gravity. The second term is the action of the gas flow which is causing deformation because of the low pressure region at the wave crest and high pressure regions in the wave trough.

Thus the mechanism for instability is the pressure caused by a Bernoulli effect becoming stronger than the restoring forces of gravity and / or surface tension.

This effect is definitely present. However, this mechanism is not as efficient as another mechanism and hence is not responsible for the initial formation of waves in air-water flows and other sytems where there is a significant viscosity difference between the phases. This general mechanism may be important for growth of *large* amplitude waves.

Orr-Sommerfeld Equation

A good reference for this section is R. L. Panton, Incompressible flow, Wiley, 1984

Here we derive the Orr-Sommerfeld equation which is a 4th order ODE that describes the growth on infinitesimal periodic distrubances that are governed by the Navier-Stokes equations.

The linearized x-momentum equations for a nearly-parallel flow are
x direction

```
xmom = ∂t u[x, y, t] + u0[y] ∂x u[x, y, t] + v[x, y, t] ∂y u0[y] +
  ∂x p[x, y, t] - (∂{x,2} u[x, y, t] + ∂{y,2} u[x, y, t]) / Re
```

$$v(x, y, t)\,\mathrm{u0}'(y) + u^{(0,0,1)}(x, y, t) + p^{(1,0,0)}(x, y, t) + \mathrm{u0}(y)\,u^{(1,0,0)}(x, y, t) - \frac{u^{(0,2,0)}(x, y, t) + u^{(2,0,0)}(x, y, t)}{\mathrm{Re}}$$

y direction

```
ymom = ∂t v[x, y, t] + u0[y] ∂x v[x, y, t] + ∂y p[x, y, t] -
  (∂{x,2} v[x, y, t] + ∂{y,2} v[x, y, t]) / Re
```

$$v^{(0,0,1)}(x, y, t) + p^{(0,1,0)}(x, y, t) + \mathrm{u0}(y)\,v^{(1,0,0)}(x, y, t) - \frac{v^{(0,2,0)}(x, y, t) + v^{(2,0,0)}(x, y, t)}{\mathrm{Re}}$$

The continuity equation is

```
cont = ∂x u[x, y, t] + ∂y v[x, y, t]
```

$$v^{(0,1,0)}(x, y, t) + u^{(1,0,0)}(x, y, t)$$

We need to expand these in terms of normal modes where the variable will be assumed to have the form ξ[x,y,t]=$\hat{\xi}$[y] Exp[I(α x -α c t)] based on the idea that we need functions that can describe spatially and temporally varying disturbances

```
os1 = xmom /. {u[x, y, t] → uh[y] Exp[I (α x - α c t)], v[x, y, t] →
    vh[y] Exp[I (α x - α c t)], p[x, y, t] → ph[y] Exp[I (α x - α c t)],
   u^(a1_,a2_,a3_) [x, y, t] :→ ∂{x,a1},{y,a2},{t,a3} (uh[y] Exp[I (α x - α c t)]),
   v^(a1_,a2_,a3_) [x, y, t] :→ ∂{x,a1},{y,a2},{t,a3} (vh[y] Exp[I (α x - α c t)]),
   p^(a1_,a2_,a3_) [x, y, t] :→ ∂{x,a1},{y,a2},{t,a3} (ph[y] Exp[I (α x - α c t)])}
```

$$i\,e^{i(x\alpha - c t\alpha)}\,\alpha\,\mathrm{ph}(y) - i\,c\,e^{i(x\alpha - c t\alpha)}\,\alpha\,\mathrm{uh}(y) + i\,e^{i(x\alpha - c t\alpha)}\,\alpha\,\mathrm{u0}(y)\,\mathrm{uh}(y) + e^{i(x\alpha - c t\alpha)}\,\mathrm{vh}(y)\,\mathrm{u0}'(y) - \frac{e^{i(x\alpha - c t\alpha)}\,\mathrm{uh}''(y) - e^{i(x\alpha - c t\alpha)}\,\alpha^2\,\mathrm{uh}(y)}{\mathrm{Re}}$$

```
os2 = Cancel[Expand[os1 / Exp[I (α x - α c t)]]]
```

$$\frac{\text{uh}(y)\,\alpha^2}{\text{Re}} + i\,\text{ph}(y)\,\alpha - i\,c\,\text{uh}(y)\,\alpha + i\,\text{u0}(y)\,\text{uh}(y)\,\alpha + \text{vh}(y)\,\text{u0}'(y) - \frac{\text{uh}''(y)}{\text{Re}}$$

Which is the result for the x equation

```
os3 = ymom /. {u[x, y, t] → uh[y] Exp[I (α x - α c t)], v[x, y, t] →
     vh[y] Exp[I (α x - α c t)], p[x, y, t] → ph[y] Exp[I (α x - α c t)],
   u^(a1_,a2_,a3_)[x, y, t] :→ ∂_{x,a1},{y,a2},{t,a3} (uh[y] Exp[I (α x - α c t)]),
   v^(a1_,a2_,a3_)[x, y, t] :→ ∂_{x,a1},{y,a2},{t,a3} (vh[y] Exp[I (α x - α c t)]),
   p^(a1_,a2_,a3_)[x, y, t] :→ ∂_{x,a1},{y,a2},{t,a3} (ph[y] Exp[I (α x - α c t)])}
```

$$-i\,c\,e^{i(x\alpha - c\,t\,\alpha)}\,\alpha\,\text{vh}(y) + i\,e^{i(x\alpha - c\,t\,\alpha)}\,\alpha\,\text{u0}(y)\,\text{vh}(y) + e^{i(x\alpha - c\,t\,\alpha)}\,\text{ph}'(y) - \frac{e^{i(x\alpha - c\,t\,\alpha)}\,\text{vh}''(y) - e^{i(x\alpha - c\,t\,\alpha)}\,\alpha^2\,\text{vh}(y)}{\text{Re}}$$

```
os4 = Cancel[Expand[os3 / Exp[I (α x - α c t)]]]
```

$$\frac{\text{vh}(y)\,\alpha^2}{\text{Re}} - i\,c\,\text{vh}(y)\,\alpha + i\,\text{u0}(y)\,\text{vh}(y)\,\alpha + \text{ph}'(y) - \frac{\text{vh}''(y)}{\text{Re}}$$

which is the result for the y equation

```
os5 = cont /. {u[x, y, t] → uh[y] Exp[I (α x - α c t)], v[x, y, t] →
     vh[y] Exp[I (α x - α c t)], p[x, y, t] → ph[y] Exp[I (α x - α c t)],
   u^(a1_,a2_,a3_)[x, y, t] :→ ∂_{x,a1},{y,a2},{t,a3} (uh[y] Exp[I (α x - α c t)]),
   v^(a1_,a2_,a3_)[x, y, t] :→ ∂_{x,a1},{y,a2},{t,a3} (vh[y] Exp[I (α x - α c t)]),
   p^(a1_,a2_,a3_)[x, y, t] :→ ∂_{x,a1},{y,a2},{t,a3} (ph[y] Exp[I (α x - α c t)])}
```

$$i\,e^{i(x\alpha - c\,t\,\alpha)}\,\alpha\,\text{uh}(y) + e^{i(x\alpha - c\,t\,\alpha)}\,\text{vh}'(y)$$

```
os6 = Cancel[Expand[os5 / Exp[I (α x - α c t)]]]
```

$$i\,\alpha\,\text{uh}(y) + \text{vh}'(y)$$

which is the continuity equation

Now we use the continuity equation and the disturbance stream function definition which is uh[y]= $\frac{\partial\phi[y]}{\partial y}$ and vh[y]=-I ϕ[y]

```
os7 = os2 /. {uh[y] → ∂y ϕ[y], vh[y] → -I α ϕ[y],
   uh(a1_)[y] :→ ∂{y,a1} (∂y ϕ[y]), vh(a1_)[y] :→ ∂{y,a1} (-I α ϕ[y])}
```

$$\frac{\phi'(y)\,\alpha^2}{\mathrm{Re}} + i\,\mathrm{ph}(y)\,\alpha - i\,\phi(y)\,\mathrm{u0}'(y)\,\alpha - i\,c\,\phi'(y)\,\alpha + i\,\mathrm{u0}(y)\,\phi'(y)\,\alpha - \frac{\phi^{(3)}(y)}{\mathrm{Re}}$$

```
os8 = os4 /. {uh[y] → ∂y ϕ[y], vh[y] → -I α ϕ[y],
   uh(a1_)[y] :→ ∂{y,a1} (∂y ϕ[y]), vh(a1_)[y] :→ ∂{y,a1} (-I α ϕ[y])}
```

$$-\frac{i\,\phi(y)\,\alpha^3}{\mathrm{Re}} - c\,\phi(y)\,\alpha^2 + \mathrm{u0}(y)\,\phi(y)\,\alpha^2 + \frac{i\,\phi''(y)\,\alpha}{\mathrm{Re}} + \mathrm{ph}'(y)$$

Now combine the x and y equations to eliminate pressure

```
os9 = ∂y os7 - I α os8
```

$$\frac{\phi''(y)\,\alpha^2}{\mathrm{Re}} + i\,\mathrm{ph}'(y)\,\alpha - i\,\phi(y)\,\mathrm{u0}''(y)\,\alpha - i\,c\,\phi''(y)\,\alpha + i\,\mathrm{u0}(y)\,\phi''(y)\,\alpha - i\left(-\frac{i\,\phi(y)\,\alpha^3}{\mathrm{Re}} - c\,\phi(y)\,\alpha^2 + \mathrm{u0}(y)\,\phi(y)\,\alpha^2 + \frac{i\,\phi''(y)\,\alpha}{\mathrm{Re}} + \mathrm{ph}'(y)\right)\alpha - \frac{\phi^{(4)}(y)}{\mathrm{Re}}$$

```
Cancel[Expand[os9 / (I α)]]
```

$$\frac{i\,\phi(y)\,\alpha^3}{\mathrm{Re}} + c\,\phi(y)\,\alpha^2 - \mathrm{u0}(y)\,\phi(y)\,\alpha^2 - \frac{2\,i\,\phi''(y)\,\alpha}{\mathrm{Re}} - \phi(y)\,\mathrm{u0}''(y) - c\,\phi''(y) + \mathrm{u0}(y)\,\phi''(y) + \frac{i\,\phi^{(4)}(y)}{\mathrm{Re}\,\alpha}$$

$$\boldsymbol{\frac{-1}{i\,\alpha\,\mathrm{Re}}\,(\phi(y)\,\alpha^4 - 2\,\phi''(y)\,\alpha^2 + \phi^{(4)}(y)) + (\phi(y)\,\alpha^2 - \phi''(y))\,(c - \mathrm{u0}(y)) - \phi(y)\,\mathrm{u0}''(y)}$$

$$-\phi(y)\,\mathrm{u0}''(y) + (c - \mathrm{u0}(y))\,(\alpha^2\,\phi(y) - \phi''(y)) + \frac{i\,(\phi(y)\,\alpha^4 - 2\,\phi''(y)\,\alpha^2 + \phi^{(4)}(y))}{\mathrm{Re}\,\alpha}$$

This, with little effort, is the Orr-Sommerfeld equation. !!!